unittest
{
	assert( [ "20 2 5" ].parse.expand.solve == 84 );
	assert( [ "10 1 2" ].parse.expand.solve == 13 );
	assert( [ "100 4 16" ].parse.expand.solve == 4554 );
}



import std.conv;
import std.functional;
import std.range;
import std.stdio;
import std.string;
import std.typecons;

void main()
{
	stdin.byLineCopy.parse.expand.solve.writeln;
}

auto parse( Range )( Range input )
if( isInputRange!Range && is( ElementType!Range == string ) )
{
	auto nab = input.front.split.to!( long[] );
	return tuple( nab[ 0 ], nab[ 1 ], nab[ 2 ] );
}

auto solve( long n, long a, long b )
{
	auto t = 0L;
	
	foreach( i; 1 .. n + 1 )
	{
		auto s = memoize!sumDigits( i );
		if( a <= s && s <= b )
		{
			t += i;
		}
	}
	
	return t;
}

long sumDigits( long i )
{
	if( i < 10 ) return i;
	else         return i % 10 + memoize!sumDigits( i / 10 );
}
